1. Si $ax+by+cz+d=0$ est ue équation d'un plan, on sait que la distance d'un point (x_0,y_0,z_0) à ce plan est $\dfrac{|ax_0+by_0+cz_0+d|}{\sqrt{a^2+b^2+c^2}}$.
2. La distance δ d'un point M à la droite $(A,\overrightarrow{u})$ se déduit de $\|\overrightarrow{MA}\wedge\overrightarrow{u}\|=\delta\|\overrightarrow{u}\|$.
   $\overrightarrow{u}$ n'est pas orthogonal à la normale à P donc la droite et le plan ne sont pas parallèles.
3. On étudie une quadrique d'équation $F(x,y,z)=0$.
   Il existe un point et un seul $\Omega=(0,-1,-1)$ critique pour F donc il s'agit d'une quadrique de centre Ω.
   On le prend comme origine d'où la nouvelle équation $S1$.
   Le premier membre peut s'écrire dans un repère orthonormé d'origine Ω :
   $\lambda_1x_1^2+\lambda_2x_2^2+\lambda_3x_3^2=5$ où les λ_i sont les valeurs propres de Q.
   Les 3 valeurs propres sont strictement positives, donc il s'agit d'un ellipsoïde, mais $\lambda_3\approx 0$ donc l'un des $1/2-$axes est très grand (c'est presque un cylindre elliptique).
   Le volume intérieur à un ellipsoïde de $1/2-$axes a,b,c est $\dfrac{4}{3}\pi abc$.

O19-C10

```
> restart;
> d1:=(x+y-2*z-1)/sqrt(6);
```

$$d1 := \frac{(x+y-2\,z-1)\sqrt{6}}{6}$$

```
> d2:=sqrt((y-2*z-1)^2+(z-x+1)^2+(2*x-y-1)^2)/sqrt(6);
```

$$d2 := \frac{\sqrt{2\,y^2-4\,y\,z+5\,z^2+6\,z+3-2\,z\,x+5\,x^2-6\,x-4\,x\,y}\,\sqrt{6}}{6}$$

```
> with(plots):
Warning, the name changecoords has been redefined
> p:=plot3d((x+y-1)/2,x=-5..5,y=-5..5):
> d:=polygonplot3d([[-1,-3,2],[3,5,-2]]):
> display3d({p,d});
```



```
> S:=expand(d1^2+d2^2=5);
```

$$S := x^2-\frac{1}{3}\,x\,y-z\,x-\frac{4}{3}\,x+\frac{1}{2}\,y^2-\frac{4}{3}\,y\,z-\frac{1}{3}\,y+\frac{3}{2}\,z^2+\frac{5}{3}\,z+\frac{2}{3}=5$$

```
> surf:=implicitplot3d(S,x=-10..10,y=-15..15,z=-12..8,numpoints=15
  000):
> display3d({d,p,surf});
```




```
> gra:={diff(S,x),diff(S,y),diff(S,z)};
```

$$gra := \{ 2\,x - \frac{y}{3} - z - \frac{4}{3} = 0,\ -\frac{x}{3} + y - \frac{4\,z}{3} - \frac{1}{3} = 0,\ -x - \frac{4\,y}{3} + 3\,z + \frac{5}{3} = 0 \}$$

```
> centre:=solve(gra);
```

$$centre := \{ x = 0,\ z = -1,\ y = -1 \}$$

```
> S1:=expand(subs({x=xx,y=-1+yy,z=-1+zz},S));
```

$$S1 := xx^2 - \frac{1}{3}\,xx\,yy - xx\,zz + \frac{1}{2}\,yy^2 - \frac{4}{3}\,yy\,zz + \frac{3}{2}\,zz^2 = 5$$

```
> with(LinearAlgebra):
> Q:=Matrix([[1,-1/6,-1/2],[-1/6,1/2,-2/3],[-1/2,-2/3,3/2]]);
```

$$Q := \begin{bmatrix} 1 & \frac{-1}{6} & \frac{-1}{2} \\ \frac{-1}{6} & \frac{1}{2} & \frac{-2}{3} \\ \frac{-1}{2} & \frac{-2}{3} & \frac{3}{2} \end{bmatrix}$$

```
> vp:=Eigenvalues(Q);evalf(%);
```

$$vp := \begin{bmatrix} 1 \\ 1 + \frac{\sqrt{35}}{6} \\ 1 - \frac{\sqrt{35}}{6} \end{bmatrix}$$

$$\begin{bmatrix} 1. \\ 1.986013297 \\ 0.0139867026 \end{bmatrix}$$

```
> volume:=4*Pi*5*sqrt(5)/3/sqrt(vp[1]*vp[2]*vp[3]);evalf(%);
```

$$volume := \frac{20\,\pi\sqrt{5}}{3\sqrt{\left(1 + \frac{\sqrt{35}}{6}\right)\left(1 - \frac{\sqrt{35}}{6}\right)}}$$

$$280.9925913$$

```
>
```